/*/{Protheus.doc} SYNCFTP
E a User Function que faz conexao com um servidor FTP e transfere o arquivo gerado do layout para a pasta definida para este layout
@author Elton Teodoro Alves
@since 14/11/2014
@version P11 R5
@Param aSyncEnv,Array,Array com os dados gerado pela user function SYNCENV
@Param cFile,Caracter,Arquivo a ser salvo no local indicado no servidor FTP
@Return Logico Indica se a funcao foi executada com sucesso
/*/
#Include 'Protheus.ch'

User Function SYNCFTP(aSyncEnv,cFile)
	
	If .Not. FtpConnect(AllTrim(aSyncEnv[11]),aSyncEnv[18],AllTrim(aSyncEnv[12]),AllTrim(aSyncEnv[13]))
		
		U_SYNCMAIL(aSyncEnv[15],aSyncEnv[19] + ' - ' + AllTrim(aSyncEnv[17]),'Erro na conexao com FTP Server. Nao foi possivel salvar arquivo : ' + cFile,AllTrim(SubStr(aSyncEnv[4],1,6)),AllTrim(SubStr(aSyncEnv[4],7,6)),1,aSyncEnv[8],aSyncEnv[9])
		
		FtpDisconnect()
		
		Return .F.
		
	EndIf
	
	If .Not. FtpUpload(cFile,If(!Empty(aSyncEnv[14]),AllTrim(aSyncEnv[14]),'/') + SubStr(cFile,At(aSyncEnv[19] + '.',cFile)))
		
		U_SYNCMAIL(aSyncEnv[15],aSyncEnv[19] + ' - ' + AllTrim(aSyncEnv[17]),'Erro ao salvar arquivo no FTP : ' + cFile,AllTrim(SubStr(aSyncEnv[4],1,6)),AllTrim(SubStr(aSyncEnv[4],7,6)),1,aSyncEnv[8],aSyncEnv[9])
		
		FtpDisconnect()
		
		Return .F.
		
	EndIf
	
	U_SYNCMAIL(aSyncEnv[15],aSyncEnv[19] + ' - ' + AllTrim(aSyncEnv[17]),'Arquivo Salvo no FTP : ' + If(!Empty(aSyncEnv[14]),AllTrim(aSyncEnv[14]),'/') + SubStr(cFile,At(aSyncEnv[19] + '.',cFile)),AllTrim(SubStr(aSyncEnv[4],1,6)),AllTrim(SubStr(aSyncEnv[4],7,6)),0,aSyncEnv[8],aSyncEnv[9])
	
	FtpDisconnect()
	
Return .T.
